

Declarative Synchronous Languages

Lustre / Signal

Languages

**Say what IS or what
SHOULD BE**

Declarative languages

Imperative languages

**Say what MUST BE
DONE**

LUSTRE/SIGNAL


Language	Characteristics	Applications
LUSTRE (and the associated tool SCADE)	Functional Mono clock	Critical systems; Command/Control; Circuits
SIGNAL	Relational (constraints) Multiple- clocked	System specification. Automatic control; Signal Data Processing

LUSTRE


- We study LUSTRE in depth because
 - It is a very simple language (4 primitive operators to express reactions)
 - Relies on models familiar to engineers
 - Equation systems
 - Data flow network
 - Lends itself to formal verification (it is a kind of logical language)
 - Very simple (mathematical) semantics

Operator Networks


- LUSTRE and SIGNAL programs can be interpreted as **networks of operators**.
- Data « flow » to operators where they are consumed. Then, the operators generate new data. (Data Flow description).



Data Flow



Functional Point of View



$$P' = P + W_N^k * Q$$

$$Q' = P - W_N^k * Q$$

LUSTRE

Flows, Clocks

- A **flow** is a pair made of
 - A possibly infinite sequence of **values** of a given type
 - A **clock** representing a sequence of instants

$X:T \quad (x_1, x_2, \dots, x_n, \dots)$

Language (1)

Variable :

- typed
- If not an input variable, defined by 1 and only 1 equation
- Predefined types: **int**, **bool**, **real**
- tuples: **(a, b, c)**

Equation : $\mathbf{x} = \mathbf{E}$ means $\forall \mathbf{k}, \mathbf{x}_{\mathbf{k}} = \mathbf{e}_{\mathbf{k}}$

Assertion :

Boolean expression that should be always **true** at each instant of its clock.

Language (2)

Substitution principle:

if $x = E$ then E can be substituted for x anywhere in the program and conversely

Definition principle:

A variable is **fully defined** by its declaration and the equation in which it appears as a left-hand side term

Expressions

Constants

`0, 1, ...`, `true, false, ...`, `1.52, ...`
`int` `bool` `real`

`+`
Imported
types and
operators

$$c : \alpha \iff \forall k \in \mathbb{N}, c_k = c$$

Operators and Types

- Strict operators:
 $\text{mod} : \text{int} \times \text{int} \rightarrow \text{int}$
- Loaded operators:
 $+ : \text{int} \times \text{int} \rightarrow \text{int}$
 $+ : \text{real} \times \text{real} \rightarrow \text{real}$
- Polymorphic operators:
 $<> : \alpha \times \alpha \rightarrow \text{bool}$

« Combinational » Lustre

Data operators

Arithmetical: `+`, `-`, `*`, `/`, `div`, `mod`

Logical: `and`, `or`, `not`, `xor`, `=>`


Conditional: `if ... then ... else ...`

Casts: `int`, `real`

« Point-wise » operators

$$X \text{ op } Y \Leftrightarrow \forall k, (X \text{ op } Y)_k = X_k \text{ op } Y_k$$

« Combinational » Example



$$\forall k \in \mathbb{N}, M_k = (X_k + Y_k) / 2_k$$

Example (suite)

```
node Average (X,Y:int)
  returns (M:int);
var S:int;  -- local variable
let
  S = X + Y;  -- non significant order
  M = S / 2;
tel
```

By **substitution**, the behavior is the same

Memorizing

Take the **past** into account!

pre (previous):

$$X = (x_1, x_2, \dots, x_n, \dots) : pre(X) = (\text{nil}, x_1, \dots, x_{n-1}, \dots)$$

Undefined value denoting uninitialized memory: **nil**

-> (initialize): sometimes call “followed by”

$$X = (x_1, x_2, \dots, x_n, \dots) , Y = (y_1, y_2, \dots, y_n, \dots) :$$

$$(X -> Y) = (x_1, y_2, \dots, y_n, \dots)$$

« Sequential » Examples

```
node Edge (X:bool) returns (E:bool);  
let  
  E = false -> X and not pre X;  
tel
```

$$E_1 = \text{false}; \forall k > 1, E_k = \overline{X_{k-1}} \wedge X_k$$

Exercice: Modify to capture possible initial edge

tuple

```
node MinMax (X:int) returns (min,max:int);  
let  
  min = X -> if (X < pre min) then X else pre min;  
  max = X -> if (X > pre max) then X else pre max;  
tel
```

Recursive definitions

Temporal recursion

Usual. Use **pre** and **->**

e.g.: $\text{nat} = 1 \rightarrow \text{pre nat} + 1$

Instantaneous recursion

e.g.: $X = 1.0 / (2.0 - X)$

Forbidden in Lustre, even if a solution exists!

Be carefull with cross-recursion.

Clocks

Basic clock

Discrete time induced by the input sequence

Derived clocks (slower)

when (filter operator):

E when C is the sub-sequence of **E** obtained by keeping only the values of indexes e_k for which $c_k = \text{true}$

Examples of clocks

Basic cycles	1	2	3	4	5	6	7	8
C1	true	false	true	true	false	true	false	true
Cycles of C1	1		2	3		4		5
C2	false		true	false		true		true
Cycles of C2			1			2		3

Example of sampling

For

nat, odd: int

halfBaseClock: bool

nat = 1 -> pre nat + 1;

halfBaseClock =

 true -> not pre halfBaseClock;

odd = nat when halfBaseClock;

nat is a flow on the basic clock;

odd is a flow on halfBaseClock

Exercise: write even

Interpolation operator

« converse » of sampling

current (interpolation) :

Let E be an expression whose clock is C , **current** (E) is an expression on the clock of C , and its value at any instant of this clock is the value of E at the last time when c was true.



current (X when C) $\neq X$

current can yield `nil`

Example of current

Basic cycles	1	2	3	4	5	6	7
C	ff	tt	ff	tt	ff	ff	tt
X	x1	x2	x3	x4	x5	x6	x7
Y = X when C		x2		x4			x7
Z = current(Y)	nil	x2	x2	x4	x4	x4	x7

Other examples of current

X	1	2	3	4	5	6	7
Y	t	f	t	t	t	f	f
C	t	t	f	t	t	f	t
Z=X when C	1	2		4	5		7
H=Y when C	t	f		t	t		f
T=Z when H	1			4	5		
current T	1	1		4	5		5
current (current T)	1	1	1	4	5	5	5

The initialization issue

$Y = \text{current}(X \text{ when } C)$ where $C_1 = \text{false}$ is erroneous.

Possible solutions:

- Strict discipline: ensure that C_1 is always true.
- Force the clock to true at the first instant:

$CC = \text{true} \rightarrow C; Y = \text{current}(X \text{ when } CC);$

- Provide a default value D :

$Y = \text{if } C \text{ then } \text{current}(X \text{ when } C) \text{ else } D \rightarrow \text{pre } Y;$

First programs

Counters

```
node COUNTER (init, incr:int; reset:bool)
    returns (n:int);
let
    n = init -> if reset then init else
                pre(n) + incr;
tel;
```

C	tt	ff	tt	tt	ff	ff	tt
COUNTER(0,2,false)	0	2	4	6	8	10	12
COUNTER((0,2,false)when C)	0		2	4			6
COUNTER(0,2,false)when C	0		4	6			12

Edges


```
node Edge (b:bool) returns (f:bool);  
-- detection of a rising edge  
let  
    f = false -> (b and not pre(b)) ;  
tel;
```

initial

Undefined at
the first instant

```
Falling_Edge = Edge(not c) ;
```

Retriggerable Monostable

```
node MR(set:bool;delay:int) returns (z:bool);  
-- retriggerable monostable  
var n: int;  Local variable  
let  
  z = ( n > 0 );  
  n = 0 -> if set then  
    delay  
  else  
    if pre(n) = 0 then  
      0  
    else  
      pre(n) - 1;  
tel;
```

2 equations:
**non
significant
order**


Non Retriggerable Monostable


```
node MNR ( e:bool; p:int )
          returns (s:bool) ;
var set:bool;
let
    s = MR (set,p) ;
    set = Edge(e) and not pre(s) ;
tel;
```

Adder

```
node ADD1 (a,b,ci:bool)
    returns (s,co:bool);

let
    s = ci xor a xor b;
    co = (b and ci) or (ci and a)
        or (a and b);
tel
```






```
node ADD4 (a0,a1,a2,a3:bool;  
           b0,b1,b2,b3:bool)  
  returns (s0,s1,s2,s3:bool;  
           carry:bool) ;  
var c0,c1,c2,c3:bool;  
let  
  (s0,c0) = ADD1(a0,b0,false) ;  
  (s1,c1) = ADD1(a1,b1,c0) ;  
  (s2,c2) = ADD1(a2,b2,c1) ;  
  (s3,c3) = ADD1(a3,b3,c2) ;  
  carry = c3 ;  
tel;
```

Verification


Principle




```
node Proof(I)
    returns (ok:bool);
var O;
let
    O=P(I);
    ok=Prop(I,O);
    assert Hypo(I,O);
tel
```

Model-checking

Enumerative algorithm. The automaton is checked state by state, starting from the initial one.



The property holds




The property is false


Model-checking (2)

Symbolic forward algorithm: The state of reachable states is built as a Boolean formula over the state variables

Post*



The property holds




The property is false

Model-checking (3)

Symbolic backward algorithm: This algorithm builds a symbolic representation of bad states.


Pre*

Initial state



The property holds

Initial state



The property is false



```

node verif_ADD (A,B:bool^4)
    returns (ok:bool) ;
var Sa,Sb:bool^4; ca,cb:bool;
let
    (Sa,ca) = ADD(4,A,B) ;
    (Sb,cb) = ADD4b(A,B) ;
    ok = (ca = cb)
    and (Sa[0]=Sb[0]) and (Sa[1]=Sb[1])
    and (Sa[2]=Sb[2]) and (Sa[3]=Sb[3]) ;

```

Safety properties

Safety property: bad things never happen.

let **P** be a safety property.

Define a Boolean expression **B** such that **P** is true iff **B** is **true** through all program executions.

```
e.g.:  node implies (A,B:bool)
        returns (AiB:bool) ;
let
    AiB = not A or B;
tel
```

```

node neverBefore (B:bool)
    returns (nB:bool) ;
let
    nB = (not B) -> (not B and pre(nB)) ;
tel

```

k	1	2	3	4	5	6
B	tt	tt	tt	tt	tt	tt
nB	ff	ff	ff	ff	ff	ff

```

node neverBefore (B:bool)
    returns (nB:bool) ;
let
    nB = (not B) -> (not B and pre(nB)) ;
tel

```

k	1	2	3	4	5	6
B	ff	ff	ff	tt	tt	tt
nB	tt	tt	tt	ff	ff	ff

```

node since (X,Y:bool)
    returns (XsY:bool) ;
let
    XsY = if Y then X else
           true -> X or pre (XsY) ;
tel

```

k	1	2	3	4	5	6
X	ff	ff	ff	tt	ff	tt
Y	ff	ff	tt	ff	ff	tt
XsY	tt	tt	ff	tt	tt	tt

Any occurrence of **A** is followed by an occurrence of **B** before the next occurrence of **C**

future

C **A** **B** **B** **C** **A** **B** **C**

time

Any time **C** occurs, either **A** never occurred previously, or **B** has occurred since the last occurrence of **A**

past

```
node unBentreAetC (A,B,C:bool)
    returns (X:bool);
```

```
let
```

```
X = implies (C, neverBefore (A)
```

```
or
```

```
since (B,A) );
```


```
tel
```

or **B** has
occurred since
the last occur-
rence of **A**

Any time **C** occurs, either
A has never occurred
previously,

Lustre

Advanced programming



```
node ADD4 (a0,a1,a2,a3:bool;  
           b0,b1,b2,b3:bool)  
  returns (s0,s1,s2,s3:bool;  
          carry:bool) ;  
var c0,c1,c2,c3:bool;  
let  
  (s0,c0) = ADD1(a0,b0,false) ;  
  (s1,c1) = ADD1(a1,b1,c0) ;  
  (s2,c2) = ADD1(a2,b2,c1) ;  
  (s3,c3) = ADD1(a3,b3,c2) ;  
  carry = c3 ;  
tel;
```

```
node ADD4a (A,B:bool^4)
  returns (S:bool^4; carry:bool);

var C:bool^4;
let
  (S[0],C[0])= ADD1 (A[0],B[0],false);
  (S[1..3],C[1..3])= ADD1 (
    A[1..3],B[1..3],C[0..2]);
  carry = C[3];
tel;
```

```

node ADD (const n; A,B:bool^n)
  returns (S:bool^n; carry:bool);
var C:bool^n;
let
  (S[0],C[0])= ADD1(A[0],B[0],false);
  (S[1..n-1],C[1..n-1])= ADD1(
    A[1..n-1],B[1..n-1],C[0..n-2]);
  carry = C[n-1];
tel;

```

```
const nbit = 4;
```

```

node Main_ADD (A,B:bool^nbit)
  returns (S:bool^nbit; carry:bool);
let
  (S,carry)= ADD(nbit,A,B);
tel;

```

Array operators

- constructor $[\cdot]$

if $E_0, E_1, \dots, E_{n-1}$ are expressions of type T
then $[E_0, E_1, \dots, E_{n-1}]$ is of type T^n

- concatenation $|$

let $X : T^n$ and $Y : T^m$

$X | Y$ is of type $T^{(n+m)}$

Mutual exclusion

$$EX[i] = \left| \left\{ j \leq i \mid X[j] = \text{true} \right\} \right| \leq 1$$

$$OR[i] = \bigvee_{j \leq i} X[j]$$

At most
1 true

At least
1 true

Recurrent equations:

$$EX[i+1] = EX[i] \wedge \neg (OR[i] \wedge X[i+1]) \quad \text{where } EX[0] = \text{true}$$

$$OR[i+1] = OR[i] \vee X[i+1] \quad \text{where } OR[0] = X[0]$$

OR[0] = X[0]

OR[1..n-1]

OR = [X[0]] | (OR[0..n-2] or X[1..n-1]);

OR[i+1] = OR[i] ∨ X[i+1]


$EX[0] = \text{true}$

$EX[1..n-1]$


$EX = [\text{true}] \mid (EX[0..n-2] \text{ and not } (OR[0..n-2] \text{ and } X[1..n-1]));$

$EX[i+1] = EX[i] \wedge \neg (OR[i] \wedge X[i+1])$

```
node exclusive (const n:int; X:bool^n)
    returns (excl:bool);
var EX, OR: bool^n;
let
    excl = EX[n-1];
    EX = [true] | (EX[0..n-2] and
                    not (OR[0..n-2] and X[1..n-1]));
    OR = [X[0]] | (OR[0..n-2] or X[1..n-1]);
tel;
```



Tools



Compilation : source code

```
node Counter(X, reset:bool) returns (C:int);
```

```
let
```

```
    C =      if reset then 0
              else      if X then (0 -> pre C) + 1
                        else (0 -> pre C) ;
```

```
tel
```

```
node Edge(X:bool) returns (F:bool);
```

```
let
```

```
    F = X -> X and not pre X;
```

```
tel
```


```
node FilteredCnt(X, reset:bool) returns (cpt:int);
```

```
let
```

```
    cpt = Counter(Edge(X), (true -> pre reset));
```

```
tel
```

Compilation : expansion (1)



Compilation : expansion (2)

node FilteredCnt(X, reset:bool) returns (cpt:int);

var

X_1, reset_1:bool; -- inputs of Counter

C_1:int; -- output of Counter

X_2, F_2:bool; -- input and output of Edge

let

-- Calling Counter:

C_1 = if reset_1 then 0

else if X_1 then (0 -> pre C_1) + 1

else (0 -> pre C_1) ;

X_1 = F_2;

reset_1 = true -> pre reset;

cpt = C_1;


-- calling Edge:

F_2 = X_2 -> X_2 and not pre X_2;

X_2 = X;

tel

Compilation : expansion (3)



Compilation : expansion(4)

node FilteredCnt(X, reset:bool) returns (cpt:int) ;

var

F, **R**:bool;

let

cpt = if **R** then 0

else if **F** then (0 -> pre cpt) + 1

else (0 -> pre cpt) ;

R = true -> pre reset;

F = X -> X and not pre X;

tel

Compilation: to imperative code

where

memories

pcpt stands for « pre cpt »

pX stands for « pre X »

preset stands for « pre reset »

```
cpt = if R then 0
      else
          if F then (if init then 0 else pcpt) + 1
          else (if init then 0 else pcpt) ;

R = if init then true else preset;
F = if init then X else (X and not pX);
```

assignments

Compilation : sequentialization


R = if init then true else preset;

F = if init then X else (X and not pX) ;

cpt = if **R** then 0

else if **F** then (if init then 0 else pcpt) + 1

else (if init then 0 else pcpt) ;



Ordered
(assignments)

Compilation : C code (1)

```
/* memories are global because their values are persistent
from a cycle to the next one */
static int pcpt;
static bool pX;
static bool preset;
static bool init = true; /* Initialization */

/* Function that implements a reaction */
void FilteredCnt_step (
    bool X; bool reset; /* inputs passed by value */
    int* cpt /* output passed by reference */
)
```

Compilation : C code (2)

```
{  
    bool R, F; /* local variables. Non persistent values */  
    /* Local variables and output computation */  
    R = init? true : preset;  
    F = init? X : (X && ! pX);  
    *cpt = R? 0 :  
        F? ((init? 0 : pcpt) + 1) :  
            init? 0 : pcpt;  
    /* Memorizing */  
    pcpt = *cpt;  
    pX = X;  
    preset = reset;  
    init = false;  
}
```

Compilation : C code (3)

```
{  
    bool F; /* local variables. Non persistent values */  
    if (init) {  
        /* simplified code when init = true; F useless */  
        *cpt = 0;  
        /* assignment executed once only */  
        init = false;  
    } else {  
        F = (X && ! pX);  
        *cpt = preset? 0 :  
            F? (pcpt + 1): pcpt;  
    }  
    /* Memorizing */  
    pcpt = *cpt;  
    pX = X;  
    preset = reset;  
}
```

**imperative
if then else**

Compilation (format)

Single loop

- size of the code **linear** w.r.t. the size of the source code
- **Systematic execution** of the body: **may be slow!**

Automaton

- Size of the code can be **exponential**
- **Execution:** depends on the current state (switch): **fast**

SIGNAL

Signal

- A **signal** is a sequence of values associated with a **clock**.
- Data:
 - Scalar types (**boolean**, **integer**, **float**)
 - Arbitrary dimension array of scalar elements
 - type **event** , which may be present or absent
- A clock is a set of discrete instants out of a totally ordered set.

Let X be a signal. The clock C_X associated with X is defined by the sequence of instants when X is present.

$$X = (X_t)_{t \in C_X}$$

Given a clock C , one defines

$$0_C = \min \{ t \mid t \in C \}$$

$$\forall t \in C, t \neq 0_C, t \ll_C 1 = \max \{ t' \in C, t' < t \}$$

$$t \ll_C k + 1 = \max \{ t' \in C, t' \ll_C k \} \text{ si } t \ll_C k \neq 0_C$$

Operators

Signals are defined by elementary processes built with two kinds of operators:

- Usual operators extended to sequences

$$Y := f(X_1, \dots, X_n) \quad Y, X_1, \dots, X_n \text{ same clock}$$

- Temporal operators:

- delay $Y := X \$ k$

$$\forall t \in C \mid t \ll_C k, \exists Y_t = X_{t \ll_C k}$$

- extraction $Y := X \text{ when } B$ (B Boolean signal)

$$C_Y = \left\{ t \in C_X \cap C_B \mid B_t = \text{true} \right\} \text{ et } \forall t \in C_Y, Y_t = X_t$$

- deterministic merge

$$Y := X \text{ default } Z \quad C_Y = C_X \cup C_Z$$

$$\forall t \in C_X \cup C_Z, \quad Y_t = \begin{cases} X_t & \text{if } t \in C_X \\ Z_t & \text{otherwise} \end{cases}$$

Consequence: Clocks in Signal are not trees (differs from Lustre)

Composition

Two composition operators:

- parallel composition

$P \mid Q$

Conjunction
of constraints

- restriction

$P \setminus X$

makes X
local to P

processus signal

+ synchro operator

Example of Signal program

The signal **MIN** must be emitted every 60 **SEC**

```
( | S := (0 when MIN) default (ZS+1)
  | ZS := S $ 1
  | MIN := SEC when (ZS=59)
  | synchro {S,SEC}
  | )
```

Clock calculus

- Relation between clocks are encoded as polynomial over the field $\mathbb{Z} / 3\mathbb{Z}$
- Absent $\leftrightarrow 0$
- Boolean present **true** $\leftrightarrow 1$
- Boolean present **false** $\leftrightarrow -1$

Clocks (continued)

Given a Boolean flow s , $-s - s^2 = 1$ iff s is present and **true**.

v when $s \Rightarrow$ product of the polynomial of v by $-s - s^2$

The polynomial $1 - s^2$ is 0 iff s is present.

s default $v \Rightarrow$ $s + (1 - s^2)v$

For non Boolean flows, all present values are represented by 1, all absent values by 0.